Sommersemester 2015

Technische Universität Ilmenau
Institut für Mathematik
PD Dr. Erhard Hexel
Mathematik mit Maple für IN

Übungsserie 6

Reelle Funktionen f: $\mathbb{R}^n$ -> $\mathbb{R}$ mit Maple

1. Graphische Darstellung und Niveaumengen im Fall n=2

> $restart : with(plots) :$

> $f := (x, y) \rightarrow -\dfrac{x}{x^2 + y^2 + 2}$

> $'f(x, y)' = f(x, y)$

Graphische Darstellung der Funktion f(x,y):

> $plot3d(f(x, y), x = -4..4, y = -4..4, axes = BOXED, style = PATCHCONTOUR, orientation = [59, 63])$

Die Niveaumengen N_c(f) = {(x,y) | f(x,y)=c} der Funktion f(x,y) sind ihre sogenannten **Höhenlinien**:

> $contourplot(f(x, y), x = -4..4, y = -4..4)$

Graphische Darstellung der Schnittkurve der von z=f(x,y) definierten Fläche im $\mathbb{R}^3$ mit der zur x-z-Ebene parallelen Ebene y=a (man aktiviere dazu die Animation):

> $'f(x, a)' = f(x, a)$

> $animate(f(x, a), x = -4..4, a = -4..4, color = red)$

2. Grenzwertbestimmung

> $restart$

Grenzwerte algebraischer mehrstelliger Funktionen können, wenn sie existieren, mit dem **limit**-Befehl berechnet werden:

> $limit\left(\dfrac{x - y}{y}, \{x = 0, y = 1\}\right)$

> $limit\left(\dfrac{x^2 + 1}{y}, \{x = 0, y = infinity\}\right)$

> $limit(x \cdot y, \{x = 0, y = infinity\})$

Die Grenzwertberechnung für mehrstellige Funktionen, die nicht algebraisch sind, ist mit Maple nicht möglich:

> $limit\left(\exp\left(\dfrac{x - y}{y}\right), \{x = 0, y = 1\}\right)$

> $limit\left(\dfrac{\sin(x)}{x + y}, \{x = 0, y = 1\}\right)$

3. Partielle Ableitungen, Gradient und Differential

Die partiellen Ableitungen einer Funktion f: $\mathbb{R}^n \to \mathbb{R}$ mit n≥2 können mit den Maple-Befehlen **D** und **diff** berechnet werden, wobei (analog zum eindimensionalen Fall) mit **D** eine Funktion und mit **diff** ein Term erzeugt wird.

```
> restart : with(plots) : with(LinearAlgebra) :
> f := (x, y) -> x^2·sqrt(y)
> plot3d(f(x, y), x = -2..2, y = 0..4, axes = BOXED, style = PATCHCONTOUR, orientation
      = [-40, 50])
> Diff( f(x, y), x) = diff( f(x, y), x)
> Diff( f(x, y), y) = diff( f(x, y), y)
> fx := D[1](f)     # entspricht dem Befehl  fx:=unapply(diff( f(x, y), x), x, y)
> fy := D[2](f)     # entspricht dem Befehl  fy:=unapply(diff( f(x, y), y), x, y)
```

Der **Gradient** von f(x,y):

```
> gradf := (x, y) -> Vector([fx(x, y), fy(x, y)])
> gradf(x, y)
```

Der Gradient von f im Punkt a=(1,1):

```
> gradf(1, 1)
```

Alternative Berechnung des Gradienten von f:

```
> with(Student[MultivariateCalculus])
> Gradient( f(x, y), [x, y])
> Gradient( f(x, y), [x, y] = [1, 1])
```

Die **Richtungsableitung** $\frac{\partial}{\partial v} f(a)$ von f im Punkt a=(1,1) in Richtung des Vektors v=(2,-5), das ist der Anstieg von f im Punkt a in v-Richtung:

```
> v := Vector([2, -5])
> v0 := Normalize(v, 2)
> Diff( f,'v')(a) = DotProduct(gradf(1, 1), v0)
```

Das **Differential** $df(x, y, dx, dy) = gradf \circ (dx, dy)$ von f **erster Ordnung**:

```
> d := Vector([dx, dy])
> df := unapply((DotProduct(gradf(x, y), d, conjugate = false)), x, y, dx, dy)
> 'df(x, y, dx, dy)' = df(x, y, dx, dy)
```

Die Gleichung der **Tangentialebene** im Punkt a=(x0,y0)=(1,1) an die Fläche z=f(x,y):

z=T(x,y)=f(x0,y0)+df(x0,y0,x-x0,y-y0)=f(x0,y0)+gradf(x0,y0)∘(x-x0,y-y0)

```
> T := unapply( f(1, 1) + df(1, 1, x - 1, y - 1), x, y)
> plot3d([ f(x, y), T(x, y)], x = -2..2, y = 0..4, axes = BOXED, style = PATCHCONTOUR,
     view = [-2..2, 0..4, -2..8], orientation = [-70, 57], color = [red, green])
```

4. Partielle Ableitungen höherer Ordnung und Hessematrix

Partielle Ableitungen zweiter und dritter Ordnung von f(x,y):

```
> Diff('f(x, y)', x, y) = diff( f(x, y), x, y)
> Diff('f(x, y)', y, x, x) = diff( f(x, y), y, x, x)
> fxy := D[1, 2](f)
> fyxx := D[2, 1, 1](f)
> fxy(x, y)
> fyxx(x, y)
```

Die Berechnung der **Hessematrix** kann mit dem **Hessian**-Befehl nach Laden des Zusatzpaketes **VectorCalculus** erfolgen:

```
> with(VectorCalculus) :
```

> $H := Hessian(f(x,y), [x,y])$

> $Hessian(f(x,y), [x,y] = [1,1])$

Das **Differential** $d2f(x,y,dx,dy) = (dx,dy) \cdot H \cdot (dx,dy)^T$ von f **zweiter Ordnung**:

> $d2f := unapply(BilinearForm(d, d, H, conjugate = false), x, y, dx, dy)$

> $'d2f(x,y,dx,dy)' = d2f(x,y,dx,dy)$

> $d2f(1,1,dx,dy)$

Taylorpolynome erster und zweiter Ordnung von f(x,y) an der Stelle a=(1,1):

> $T1 := unapply(f(1,1) + df(1,1,x-1,y-1), x, y)$

> $T2 := unapply\left(f(1,1) + df(1,1,x-1,y-1) + \frac{1}{2} \cdot d2f(1,1,x-1,y-1), x, y\right)$

Alternative Berechnung der Taylorpolynome T1 und T2 mit dem **mtaylor**-Befehl:

> $mtaylor(f(x,y), [x=1, y=1], 2)$

> $mtaylor(f(x,y), [x=1, y=1], 3)$

Aufgaben

Aufgabe 1

Es sei z=f(x,y) die Funktion der Fläche im $\mathbb{R}^3$, die bei Rotation der Kurve z=g(x) mit x≥0 um die z-Achse entsteht. Stellen Sie f(x,y) und ihre Höhenlinien mit Maple graphisch dar für

a) z=g(x)=$\frac{\sin^2(x)}{4}$

b) z=g(x)=$e^{-x} + 2$

Aufgabe 2

Berechnen Sie mit Maple (falls möglich) die Grenzwerte der Funktion

$$f(x,y)=\frac{x^2 - y^2}{x^2 + y^2}$$

an den folgenden Stellen:

a) (x,y)=(0,1)
b) (x,y)=(1,-1)

Aufgabe 3

Gegeben sei die Funktion $f(x,y) = \frac{\sqrt{4 - x^2 + y^2}}{2}$.

Berechnen Sie mit Hilfe von Maple:
a) alle partiellen Ableitungen (erster Ordnung) von f
b) den Gradient von f im Punkt a=(1/2,-7/2)
c) die Richtungsableitung von f im Punkt a in Richtung des Vektors v=(1,-1)
d) das Differential von f
e) die Gleichung der Tangentialebene z=T(x,y) an die Fläche z=f(x,y) im Punkt (x,y)=(1,$\sqrt{2}$)

Stellen Sie mit Hilfe von Maple graphisch dar:

f) z=f(x,y) und z=T(x,y) im Punkt (x0,y0)=(1,$\sqrt{2}$) in einem gemeinsamen Bild
g) die Höhenlinien von f(x,y)

Aufgabe 4

Bestimmen Sie mit Hilfe von Maple das Differential df=df(x,y,z,dx,dy,dz) der Funktion

$$f(x, y, z) = \ln\left(\sqrt{(2x^2 + 2y^2 + 2z^2)^3}\right)$$

an der Stelle (x,y,z)=(-1,2,-2). Welchen Funktionswert liefert das Differential df näherungsweise für (-0.9,1.8,-2.1)?

Aufgabe 5

Gegeben sei die von einem Parameter C abhängige Fläche z=f(x,y) im $\mathbb{R}^3$ durch

$$f(x, y) = 89\,x^2 - 96\,xy + 61\,y^2 - 260\,x + 70\,y + C.$$

Berechnen Sie mit Hilfe von Maple den Parameter C so, dass die Fläche z=f(x,y) die x-y-Ebene berührt. Welches sind die Koordinaten des Berührungspunktes?

Aufgabe 6

Bestimmen Sie mit Hilfe von Maple den Gradienten und die Hessematrix der Funktion

$$f(x, y) = (x - y) \cdot \sin(y).$$

Bestimmen Sie weiterhin die Differentiale erster und zweiter Ordnung sowie das Taylorpolynom zweiter Ordnung von f(x,y) an der Stelle (x0,y0)=(1,1).